

Building projects with CMake

Victor Eijkhout

Fall 2025

Justification

CMake is a portable build system that is becoming a *de facto* standard for C++ package management.

Also usable with C and Fortran.

Table of contents

Help! This software uses CMake!

- Using a cmake-based library

- Using CMake packages through pkgconfig

Help! I want to write CMake myself!

- Make your own CMake configuration

Help! I want people to use my CMake package!

- Making your package discoverable through pkgconfig

Example libraries

- Parallelism

More

- Data packages

- More libraries

Help! This software uses CMake!

Using a cmake-based library

What are we talking here?

- ▶ You have downloaded a library
- ▶ It contains a file `CMakeLists.txt`
- ▶ \Rightarrow you need to install it with CMake.
- ▶ ... and then figure out how to use it in your code.

Building with CMake

- ▶ Use CMake for the configure stage, then make:

```
1 cmake -D CMAKE_INSTALL_PREFIX=/home/yourname/packages \  
2       /home/your/software/package ## source location  
3 make  
4 make install
```

or

- ▶ do everything with CMake:

```
1 cmake ## arguments  
2 cmake --build ## stuff  
3 cmake --install ## stuff
```

We focus on the first option; the second one is portable to non-Unix environments.

What does this buy you?

1. The source directory is untouched
2. The build directory contains all temporaries
3. Your install directory (as specified to CMake) now contains executables, libraries, headers etc.

You can add these to `$PATH`, compiler options, `$LD_LIBRARY_PATH`.

But see later ...



The build/make cycle

CMake creates makefiles;
makefiles ensure minimal required compilation

```
1  cmake          ## make the makefiles
2  make           ## compile your project
3  emacs onefile.c ## edit
4  make           ## minimal recompile
5
```

Only if you add (include) files do you rerun CMake.

Directory structure: two options



- ▶ In-source build: pretty common
- ▶ Out-of-source build: cleaner because never touches the source tree
- ▶ Some people skip the install step, and use everything from the build directory.

Out-of-source build: preferred

- ▶ Work from a build directory
- ▶ Specify prefix and location of CMakeLists.txt

```
1 ls some_package_1.0.0                # we are outside the source
2 ls some_package_1.0.0/CMakeLists.txt # source contains cmake file
3 mkdir builddir                        # location for temporaries
4 cd builddir                          # goto build location
5 cmake -D CMAKE_INSTALL_PREFIX=../installdir \
6     ../some_package_1.0.0            # cmake invocation
7 make                                 # make all tmp data in build loc
8 make install                         # move only final products
```

Example: eigen

Download from

<https://eigen.tuxfamily.org/index.php>

and install.

What compiler is it finding? If you are at TACC, is it the module you have loaded?

Basic customizations

Compiler settings:

```
1 cmake -D CMAKE_CXX_COMPILER=icpx
```

Alternatively:

```
1 export CXX=icpx
2 cmake .....
```

Many settings can be done on the commandline:

```
1 -D BUILD_SHARED_LIBS=ON
```

Also check out the `ccmake` utility.

Tracing and logging

- ▶ CMake prints some sort of progress messages.
- ▶ To see commandlines:

```
1 cmake -D CMAKE_VERBOSE_MAKEFILE=ON ...  
2 make V=1
```

- ▶ CMake leaves behind a log and error file, but these are insufficient:
- ▶ ⇒ use the above verbose mode and capture all output.

Using CMake packages through pkgconfig

What are we talking here?

You have just installed a CMake-based library.
Now you need it in your own code, or in another library.
How easy can we make that?

Problem

You want to install an application/package
... which needs 2 or 3 other packages.

```
1 gcc -o myprogram myprogram.c \  
2   -I/users/my/package1/include \  
3   -L/users/my/package1/lib \  
4   -I/users/my/package2/include/package \  
5   -L/users/my/package2/lib64
```

or:

```
1 cmake \  
2   -D PACKAGE1_INC=/users/my/package1/include \  
3   -D PACKAGE1_LIB=/users/my/package1/lib \  
4   -D PACKAGE2_INC=/users/my/package2/include/package \  
5   -D PACKAGE2_LIB=/users/my/package2/lib64 \  
6   ../newpackage
```

Can this be made simpler?

Finding packages with 'pkg config'

- ▶ Many packages come with a `package.pc` file
- ▶ Add that location to `PKG_CONFIG_PATH`
- ▶ The package can now be found by other CMake-based packages.

Package config settings

Let's say you've installed a library with CMake.
Somewhere in the installation is a `.pc` file:

```
1 find $TACC_SMTHNG_DIR -name \*.pc
2 ${TACC_SMTHNG_DIR}/share/pkgconfig/smithng3.pc
```

That location needs to be on the `PKG_CONFIG_PATH`:

```
1 export PKG_CONFIG_PATH=${TACC_SMTHNG_DIR}/share/pkgconfig:${PKG_CONFIG_PATH}
```

Example: eigen

Can you find the `.pc` file in the Eigen installation?

Scenario 1: finding without cmake

Packages with a `.pc` file can be found through the `pkg-config` command:

```
1 gcc -o myprogram myprogram.c \  
2     $( pkg-config --cflags package1 ) \  
3     $( pkg-config --libs package1 )
```

In a makefile:

```
1 CFLAGS = -g -O2 $$ ( pkg-config --cflags package1 )
```

Example: eigen

Make a C++ program (extension *cpp* or *cxx*):

```
1 #include "Eigen/Core"
2 int main(int argc, char **argv) {
3     return 0;
4 }
```

Can you compile this on the commandline, using *pkg-config*? Small problem: 'eigen' wants to be called 'eigen3'.

Scenario 2: finding from CMake

You are installing a CMake-based library
and it needs Eigen, which is also CMake-based

1. you install Eigen with CMake, as above
2. you add the location of `eigen.pc` to `PKG_CONFIG_PATH`
3. you run the installation of the higher library:
this works because it can now find Eigen.

Lifting the veil

So how does a CMake install find libraries such as Eigen?

Full *CMakeLists.txt* file:

```
1 cmake_minimum_required( VERSION 3.13 )
2 project( eigentest )
3
4 find_package( PkgConfig REQUIRED )
5 pkg_check_modules( EIGEN REQUIRED eigen3 )
6
7 add_executable( eigentest eigentest.cxx )
8 target_include_directories(
9     eigentest PUBLIC
10     ${EIGEN_INCLUDE_DIRS})
```

Note 1: header-only so no library, otherwise `PACKAGE_LIBRARY_DIRS` and `PACKAGE_LIBRARIES` defined.

Note 2: you will learn how to write these configurations in the second part.

Summary for now

- ▶ You can use CMake to install libraries;
- ▶ You can use these libraries from commandline / makefile;
- ▶ You can let other CMake-based libraries find them.

Other discovery mechanisms

Some packages come with `FindWhatever.cmake` or similar files.

Add package root to `CMAKE_MODULE_PATH`

Pity that there is not just one standard.

These define some macros, but you need to read the docs to see which.

Pity that there is not just one standard.

Some examples follow.

Help! I want to write CMake myself!

Make your own CMake configuration

What are we talking here?

You have a code that you want to distribute in source form for easy installation.

You decide to use CMake for portability.

You think that using CMake might make life easier.

⇒ To do: write the `CMakeLists.txt` file.

The CMakeLists file

```
1 cmake_minimum_required( VERSION 3.12 )
2 project( myproject VERSION 1.0 )
```

- ▶ Which cmake version is needed for this file?
(CMake has undergone quite some evolution!)
- ▶ Give a name to your project.
- ▶ Maybe pick a language.
C and C++ available by default, or:

```
1 enable_language(Fortran)
```

(list: *C, CXX, CSharp, CUDA, OBJC, OBJCXX, Fortran, HIP, ISPC, Swift,*
and a couple of variants of *ASM*)

Target philosophy

- ▶ Declare a target: something that needs to be built, and specify what is needed for it

```
1 add_executable( myprogram )  
2 target_sources( myprogram PRIVATE program.cxx )
```

Use of macros:

```
1 add_executable( ${PROJECT_NAME} )
```

- ▶ Do things with the target, for instance state where it is to be installed:

```
1 install( TARGETS myprogram DESTINATION . )
```

relative to the prefix location.

Example: single source

Build an executable from a single source file:

```
1 cmake_minimum_required( VERSION 3.13 )
2 project( singleprogram VERSION 1.0 )
3
4 add_executable( program )
5 target_sources( program PRIVATE program.cxx )
6 install( TARGETS program DESTINATION . )
```


Deprecated usage

Possible usage, but deprecated:

```
1 add_executable( myprogram myprogram.c myprogram.h )
```

As much as possible use 'target' design:

```
1 add_executable( program )  
2 target_sources( program PRIVATE program.cxx )
```

Exercise

- ▶ Write a 'hello world' program;
- ▶ Make a CMake setup to compile and install it;
- ▶ Test it all.

Exercise: using the Eigen library

This is a short program using Eigen:

```
1 #include <iostream>
2 #include <Eigen/Dense>
3
4 int main() {
5     // Define a 3x3 matrix
6     Eigen::Matrix3d matrix;
7     matrix << 1, 2, 3,
8               4, 5, 6,
9               7, 8, 9;
10    std::cout << "Original matrix:\n" << matrix << std::endl;
11    return 0;
12 }
```

- ▶ Make a CMake setup to compile and install it;
- ▶ Test it.

Make your own library

First a library that goes into the executable:

```
1 add_library( auxlib )  
2 target_sources( auxlib PRIVATE aux.cxx aux.h )  
3 target_link_libraries( program PRIVATE auxlib )
```

Library during build, setup

Full configuration for an executable that uses a library:

```
1 cmake_minimum_required( VERSION 3.13 )
2 project( cmakeprogram VERSION 1.0 )
3
4 add_executable( program )
5 target_sources( program PRIVATE program.cxx )
6
7 add_library( auxlib )
8 target_sources( auxlib PRIVATE aux.cxx aux.h )
9
10 target_link_libraries( program PRIVATE auxlib )
11
12 install( TARGETS program DESTINATION . )
```

Library shared by default; see later.

Shared and static libraries

In the configuration file:

```
1 add_library( auxlib STATIC )  
2 # or  
3 add_library( auxlib SHARED )
```

(default shared if left out), or by adding a runtime flag

```
1 cmake -D BUILD_SHARED_LIBS=TRUE
```

Build both by having two lines, one for shared, one for static.

Related: the `-fPIC` compile option is set by `CMAKE_POSITION_INDEPENDENT_CODE`:

```
1 cmake -D CMAKE_POSITION_INDEPENDENT_CODE=ON
```

Release a library

To have the library released too, use **PUBLIC**.

Add the library target to the **install** command.

Example: released library

```
1 cmake_minimum_required( VERSION 3.13 )
2 project( cmakeprogram VERSION 1.0 )
3
4 add_executable( program )
5 target_sources( program PRIVATE program.cxx )
6
7 add_library( auxlib STATIC )
8 target_sources( auxlib PRIVATE lib/aux.cxx lib/aux.h )
9
10 target_link_libraries( program PUBLIC auxlib )
11 target_include_directories( program PRIVATE lib )
12
13 install( TARGETS program DESTINATION bin )
14 install( TARGETS auxlib DESTINATION lib )
15 install( FILES lib/aux.h DESTINATION include )
```

Note the separate destination directories.

We are getting realistic

The previous setup was messy
Better handle the library through a recursive cmake
and make the usual `lib include bin` setup

Recursive setup, main directory

Declare that there is a directory to do recursive make:

```
1 cmake_minimum_required( VERSION 3.13 )
2 # needs >3.12 to let the executable target find the .h file
3 project( cmakeprogram VERSION 1.0 )
4
5 add_executable( program )
6 target_sources( program PRIVATE program.cxx )
7 add_subdirectory( lib )
8 target_include_directories(
9     program PUBLIC lib )
10 target_link_libraries( program PUBLIC auxlib )
11 install( TARGETS program DESTINATION bin )
```

(Note that the name of the library comes from the subdirectory)

Recursive setup, subdirectory

Installs into lib and include

```
1 cmake_minimum_required( VERSION 3.13 )
2 # needs >3.12 to let the executable target find the .h file
3
4 add_library( auxlib STATIC )
5 target_sources( auxlib
6     PRIVATE aux.cxx
7     PUBLIC aux.h )
8 install( TARGETS auxlib DESTINATION lib )
9 install( FILES aux.h DESTINATION include )
```

External libraries

- ▶ Use `LD_LIBRARY_PATH`, or
- ▶ use `rpath`.

(Apple note: forced to use second option)

```
1 set_target_properties(  
2     ${PROGRAM_NAME} PROPERTIES  
3     BUILD_RPATH    "${CATCH2_LIBRARY_DIRS};${FMTLIB_LIBRARY_DIRS}"  
4     INSTALL_RPATH "${CATCH2_LIBRARY_DIRS};${FMTLIB_LIBRARY_DIRS}"  
5 )
```

Fetch content

Include libraries actual **in** your project:

- ▶ Use the `FetchContent` module
- ▶ Declare library with `FetchContent_Declare`, build with `FetchContent_MakeAvailable`

```
1 cmake_minimum_required( VERSION 3.20 )
2 project( program VERSION 1.0 )
3
4 include( FetchContent )
5 FetchContent_Declare(
6     fmtlib
7     GIT_REPOSITORY https://github.com/fmtlib/fmt.git
8 )
9 FetchContent_MakeAvailable( fmtlib )
10
11 add_executable( program program.cxx )
12 target_link_libraries( program PRIVATE fmt::fmt )
13
14 install( TARGETS program DESTINATION . )
```

Flexibly fetching

Only fetch if needed:

- ▶ Try to find a package with *QUIET*
- ▶ Test *MYPACKAGE_FOUND*
- ▶ If not, fetch

```
1 cmake_minimum_required( VERSION 3.20 )
2 project( program VERSION 1.0 )
3
4 find_package( fmt QUIET )
5 if( fmt_FOUND )
6     message( STATUS "Found installation of fmtlib" )
7 else()
8     message( STATUS "Installing fmtlib for you" )
9     include( FetchContent )
10    FetchContent_Declare(
11        fmtlib
12        GIT_REPOSITORY https://github.com/fmtlib/fmt.git
13    )
14    FetchContent_MakeAvailable( fmtlib )
15 endif()
16
17 add_executable( program program.cxx )
18 target_link_libraries( program PRIVATE fmt::fmt )
19
20 install( TARGETS program DESTINATION . )
```

Install other project

```
1 include(ExternalProject)
2 ExternalProject_Add(googletest
3   GIT_REPOSITORY    https://github.com/google/googletest.git
4   GIT_TAG           master
5   SOURCE_DIR        "${CMAKE_BINARY_DIR}/googletest-src"
6   BINARY_DIR         "${CMAKE_BINARY_DIR}/googletest-build"
7   CONFIGURE_COMMAND ""
8   BUILD_COMMAND      ""
9   INSTALL_COMMAND    ""
10  TEST_COMMAND        ""
11 )
```

Help! I want people to use my CMake package!

Making your package discoverable through pkgconfig

How does pkgconfig work?

Use the PKG_CONFIG_PATH variable:

```
1 $ module show cxxopts 2>&1 | grep -i pkg  
2 prepend_path("PKG_CONFIG_PATH", "/opt/cxxopts/intel23/lib64/pkgconfig")
```

Write your own .pc file

configure_file line in CMakeLists.txt:

```
1 configure_file(  
2     ${CMAKE_CURRENT_SOURCE_DIR}/${PROJECT_NAME}.pc.in  
3     ${CMAKE_CURRENT_BINARY_DIR}/${PROJECT_NAME}.pc  
4     @ONLY)
```

Write your own .pc file'

The .pc.in file:

```
1 prefix="@CMAKE_INSTALL_PREFIX@"
2 exec_prefix="${prefix}"
3 libdir="${prefix}/lib"
4 includedir="${prefix}/include"
5
6 Name: @PROJECT_NAME@
7 Description: @CMAKE_PROJECT_DESCRIPTION@
8 Version: @PROJECT_VERSION@
9 Cflags: -I${includedir}
10 Libs: -L${libdir} -l@libtarget@
```

Note the initial cap!

Combination of built-in variables and your own:

```
1 set( libtarget auxlib )
```

Installing the pc file

```
1 install(  
2     FILES ${CMAKE_CURRENT_BINARY_DIR}/${PROJECT_NAME}.pc  
3     DESTINATION share/pkgconfig  
4 )
```

Example libraries

Parallelism

MPI from C

MPI has a module:

```
1 find_package( MPI )
2 target_include_directories(
3     ${PROJECT_NAME} PUBLIC
4     ${MPI_C_INCLUDE_DIRS} )
5 target_link_libraries(
6     ${PROJECT_NAME} PUBLIC
7     ${MPI_C_LIBRARIES} )
```


MPI from C++

```
1 find_package( MPI )
2 target_include_directories(
3     ${PROJECT_NAME} PUBLIC
4     ${MPI_CXX_INCLUDE_DIRS} )
5 target_link_libraries(
6     ${PROJECT_NAME} PUBLIC
7     ${MPI_CXX_LIBRARIES} )
```

MPI from Fortran90

```
1 find_package( MPI )
2 target_include_directories(
3     ${PROJECT_NAME} PUBLIC
4     ${MPI_INCLUDE_DIRS} )
5 target_link_directories(
6     ${PROJECT_NAME} PUBLIC
7     ${MPI_LIBRARY_DIRS} )
8 target_link_libraries(
9     ${PROJECT_NAME} PUBLIC
10    ${MPI_Fortran_LIBRARIES} )
```

MPI from Fortran2008

```
1 if( MPI_Fortran_HAVE_F08_MODULE )  
2 else()  
3   message( FATAL_ERROR "No f08 module for this MPI" )  
4 endif()
```

MPL

```
1 find_package( mpl REQUIRED )
2 target_include_directories(
3     ${PROJECT_NAME} PUBLIC
4     ${CMAKE_CURRENT_SOURCE_DIR}
5     mpl::mpl )
6 target_link_libraries(
7     ${PROJECT_NAME} PUBLIC
8     mpl::mpl )
```

OpenMP from C

```
1 find_package(OpenMP)  
2 target_link_libraries(  
3     ${PROJECT_NAME}  
4     PUBLIC OpenMP::OpenMP_C )
```

OpenMP from C++

```
1 find_package(OpenMP)
2 if(OpenMP_CXX_FOUND)
3 else()
4     message( FATAL_ERROR "Could not find OpenMP" )
5 endif()
6 target_link_libraries(
7     ${PROJECT_NAME}
8     PUBLIC OpenMP::OpenMP_CXX )
```

OpenMP from Fortran

```
1 enable_language(Fortran)  
2 find_package(OpenMP)  
3 target_link_libraries(  
4     ${PROJECT_NAME}  
5     PUBLIC OpenMP::OpenMP_Fortran )
```

More

TBB

```
1 find_package(TBB REQUIRED)
2 target_link_libraries( ${PROJECT_NAME} PUBLIC TBB::tbb)
```

CUDA driver

```
1 cmake_minimum_required(VERSION 3.13 FATAL_ERROR)
2 project(cmake_and_cuda)
3
4 enable_language(CUDA)
5 if ( NOT DEFINED CMAKE_CUDA_ARCHITECTURES )
6     set ( CMAKE_CUDA_ARCHITECTURES 70 )
7 endif()
8
9 add_executable(main main.cpp)
10 add_subdirectory(kernels)
11
12 # set_property(TARGET main
13 #              PROPERTY CUDA_SEPARABLE_COMPILATION ON)
14 target_link_libraries(main kernels)
15
16 install( TARGETS main DESTINATION . )
```

CUDA kernels

```
1 add_library(kernels
2     test.cu
3     test.h
4 )
5 target_compile_features(kernels PUBLIC cxx_std_11)
6 set_target_properties(
7     kernels
8     PROPERTIES CUDA_SEPARABLE_COMPILATION ON )
9 target_link_libraries(kernels)
```

Kokkos

```
1 find_package(Kokkos REQUIRED)
2 target_link_libraries(myTarget Kokkos::kokkos)
```

Either set `CMAKE_PREFIX_PATH` or add

```
1 -DKokkos_ROOT=<Kokkos Install Directory>/lib64/cmake/Kokkos
```

Maybe:

```
1 -DCMAKE_CXX_COMPILER=<Kokkos Install Directory>/bin/
   nvcc_wrapper
```

See <https://kokkos.org/kokkos-core-wiki/ProgrammingGuide/Compiling.html>

Data packages

Hdf5

C:

```
1 find_package( PkgConfig REQUIRED )
2 pkg_check_modules( HDF REQUIRED hdf5 )
3 message( STATUS "Hdf5 includes: ${HDF_INCLUDE_DIRS}" )
4
5 target_include_directories(
6     ${PROJECTNAME} PUBLIC
7     ${HDF_INCLUDE_DIRS}
8 )
```

Netcdf

C:

```
1 find_package( PkgConfig REQUIRED )
2 pkg_check_modules( NETCDF REQUIRED netcdf )
3
4 target_include_directories(
5     ${PROJECTNAME} PUBLIC
6     ${NETCDF_INCLUDE_DIRS} )
7 target_link_libraries(
8     ${PROJECTNAME} PUBLIC
9     ${NETCDF_LIBRARIES} )
10 target_link_directories(
11     ${PROJECTNAME} PUBLIC
12     ${NETCDF_LIBRARY_DIRS} )
13 target_link_libraries(
14     ${PROJECTNAME} PUBLIC netcdf )
```

Hdf5, Fortran

```
1 find_package( PkgConfig REQUIRED )
2 pkg_check_modules( NETCDF REQUIRED netcdf-fortran )
3 pkg_check_modules( NETCDF REQUIRED netcdf )
4
5 target_include_directories(
6     ${PROJECTNAME} PUBLIC
7     ${NETCDF_INCLUDE_DIRS}
8 )
9 target_link_libraries(
10     ${PROJECTNAME} PUBLIC
11     ${NETCDF_LIBRARIES} ${NETCDF_LIBRARIES}
12 )
13 target_link_directories(
14     ${PROJECTNAME} PUBLIC
15     ${NETCDF_LIBRARY_DIRS} ${NETCDF_LIBRARY_DIRS}
16 )
17 target_link_libraries(
18     ${PROJECTNAME} PUBLIC netcdf )
```


HighFive

Third party C++ interface to hdf5

```
1 find_package( HighFive REQUIRED )  
2 target_link_libraries( ${PROJECTNAME} HighFive)
```

More libraries

Package finding

Package dependent:

- ▶ Sometimes through *pkg-config*:
find the *.pc* file
- ▶ Sometimes through a *Find...* module
see CMake documentation

Catch2

```
1 find_package( PkgConfig REQUIRED )
2 pkg_check_modules( CATCH2 REQUIRED catch2-with-main )
3 target_include_directories(
4     ${PROGRAM_NAME} PUBLIC
5     ${CATCH2_INCLUDE_DIRS}
6 )
7 target_link_directories(
8     ${PROGRAM_NAME} PUBLIC
9     ${CATCH2_LIBRARY_DIRS}
10 )
11 target_link_libraries(
12     ${PROGRAM_NAME} PUBLIC
13     ${CATCH2_LIBRARIES}
14 )
```

Cxxopts

Header-only:

```
1 find_package( PkgConfig REQUIRED )
2 pkg_check_modules( OPTS REQUIRED cxxopts )
3 target_include_directories(
4     ${PROGRAM_NAME} PUBLIC
5     ${OPTS_INCLUDE_DIRS}
6 )
```

Eigen

Header-only:

```
1 cmake_minimum_required( VERSION 3.12 )
2 project( eigentest )
3
4 find_package( PkgConfig REQUIRED )
5 pkg_check_modules( EIGEN REQUIRED eigen3 )
6
7 add_executable( eigentest )
8 target_sources( eigentest PRIVATE eigentest.cxx )
9 target_include_directories(
10     myprogram PUBLIC
11     ${EIGEN_INCLUDE_DIRS})
```

Fmtlib

```
1 find_package( PkgConfig REQUIRED )
2 pkg_check_modules( FMTLIB REQUIRED fmt )
3 target_include_directories(
4     ${PROGRAM_NAME} PUBLIC ${FMTLIB_INCLUDE_DIRS}
5 )
6 target_link_directories(
7     ${PROGRAM_NAME} PUBLIC ${FMTLIB_LIBRARY_DIRS}
8 )
9 target_link_libraries(
10     ${PROGRAM_NAME} PUBLIC ${FMTLIB_LIBRARIES}
11 )
```

Range-v3

Has its own module:

```
1 find_package( range-v3 REQUIRED )
2 target_link_libraries(
3     ${PROGRAM_NAME} PUBLIC range-v3::range-v3 )
```